

## Abstract

Arrows were introduced in John Hughes paper as a general interface for computation and therefore as an alternative to monads for API design [1]. In the paper Hughes describes how arrows are a generalization of monads and how they are not as restrictive. In this paper we will use this concept to express parallelism.

First, we give an introduction to some of the possible ways to add parallelism to Haskell programs. Then, we give the basic definition of Arrows, which is followed up by the introduction of some utility functions used in this paper. Next, we introduce the `ArrowParallel` typeclass together with backends for it written with the parallel Haskells introduced earlier, finishing up with the benefits of this new way of writing parallel programs. After this we give the definition of several parallel skeletons. Then, we introduce some syntactic sugar to mimic the sequential arrow combinators introduced by Hughes [1] but with parallelism added. We also give benchmarks of our newly created parallel Haskell. Finally we give a short conclusion of what we managed to achieve.

# Contents

<b>1</b>	<b>Motivation</b>	<b>3</b>
<b>2</b>	<b>Short introduction to parallel Haskells</b>	<b>3</b>
2.1	Multicore Haskell . . . . .	3
2.2	ParMonad . . . . .	4
2.3	Eden . . . . .	5
<b>3</b>	<b>Arrows</b>	<b>6</b>
<b>4</b>	<b>Utility Functions</b>	<b>9</b>
<b>5</b>	<b>Parallel Arrows</b>	<b>10</b>
5.1	The ArrowParallel typeclass . . . . .	10
5.2	Multicore Haskell . . . . .	11
5.3	ParMonad . . . . .	11
5.4	Eden . . . . .	11
5.5	Benefits of parallel Arrows . . . . .	12
<b>6</b>	<b>Basic Skeletons</b>	<b>13</b>
6.1	parEvalNLazy . . . . .	13
6.2	parEval2 . . . . .	13
<b>7</b>	<b>Syntactic Sugar</b>	<b>15</b>
<b>8</b>	<b>Futures</b>	<b>16</b>
<b>9</b>	<b>Map-based Skeletons</b>	<b>18</b>
9.1	parMap . . . . .	18
9.2	parMapStream . . . . .	18
9.3	farm . . . . .	19
9.4	farmChunk . . . . .	19
9.5	parMapReduce . . . . .	20
<b>10</b>	<b>Topology Skeletons</b>	<b>21</b>
10.1	pipe . . . . .	21
10.2	ring . . . . .	22
10.3	torus . . . . .	23
<b>11</b>	<b>Benchmarks</b>	<b>25</b>
<b>12</b>	<b>Conclusion</b>	<b>28</b>

# 1 Motivation


Arrows were introduced in John Hughes paper as a general interface for computation and therefore as an alternative to monads for API design [1]. In the paper Hughes describes how arrows are a generalization of monads and how they are not as restrictive. In this paper we will use this concept to express parallelism.

## 2 Short introduction to parallel Haskell

There are already several ways to write parallel programs in Haskell. As we will base our parallel arrows on existing parallel Haskell, we will now give a short introduction to the ones we use as backends in this paper.

In its purest form, parallel computation (on functions) can be looked at as the execution of some functions  $a \rightarrow b$  in parallel:

```
1 parEvalN :: [a -> b] -> [a] -> [b]
```




We will now implement `parEvalN` with the different parallel Haskell.

### 2.1 Multicore Haskell

Multicore Haskell [2] is the GHC native way to do parallel processing. It ships with parallel evaluation strategies for several types which can be applied with `using :: a -> Strategy a -> a`. For `parEvalN` this means that we can just apply the list of functions  $[a \rightarrow b]$  to the list of inputs  $[a]$  by zipping them with the application operator `$`. This lazy list  $[b]$  is then forcibly evaluated in parallel with the strategy `Strategy [b]` by the `using` operator. This strategy can be constructed with `parList :: Strategy a -> Strategy [a]` and `rdeepseq :: NFData a => Strategy a`.

```
1 parEvalN :: (NFData b) => [a -> b] -> [a] -> [b]
2 parEvalN fs as = zipWith ($) fs as `using` parList rdeepseq
```



## 2.2 ParMonad


The `Par` monad introduced by Marlow et al. [3], which can be found in the `monad-par` package on hackage [4], is a monad designed for composition of parallel programs.

Our parallel evaluation function `parEvalN` can be defined by zipping the list of `[a -> b]` with the list of inputs `[a]` with the application operator `$` just like with Multicore Haskell. Then, we map over this not yet evaluated lazy list of results `[b]` with `spawnP :: NFData a => a -> Par (IVar a)` to transform them to a list of not yet evaluated forked away computations `[Par (IVar b)]`, which we convert to `Par [IVar b]` with `sequenceA`. We wait for the computations to finish by mapping over the `IVar b`'s inside the `Par` monad with `get`. This results in `Par [b]`. We finally execute this process with `runPar` to finally get `[b]` again.

```

1 parEvalN :: (NFData b) => [a -> b] -> [a] -> [b]
2 parEvalN fs as = runPar $
3   (sequenceA $ map (spawnP) $ zipWith ($) fs as) >>= mapM get

```



## 2.3 Eden

Eden is a parallel Haskell for distributed memory and comes with a MPI and a PVM backend [5–7]. This means that it works on clusters as well so it makes sense to have a Eden-based backend for our new parallel Haskell.

While it also comes with a monad PA for parallel evaluation, it also ships with a completely functional interface that includes


`spawnF :: (Trans a, Trans b) => [a -> b] -> [a] -> [b]`.

This allows us to define `parEvalN` quite easily:

```

1 parEvalN :: (Trans a, Trans b) => [a -> b] -> [a] -> [b]
2 parEvalN fs as = spawnF fs as

```



### 3 Arrows




Arrows were introduced by Hughes [1] as a general interface for computation. An arrow `arr a b` can be look at as a computation that converts an input `a` to an output `b`. This is defined in the arrow typeclass:

```

1 class Arrow arr where
2   arr :: (a -> b) -> arr a b
3
4
5
6   (>>>) :: arr a b -> arr b c -> arr a c
7
8
9
10
11  first :: arr a b -> arr (a,c) (b,c)

```



`arr` is used to lift an ordinary function to an arrow type. This can be thought of as analogous to the monadic `return`. The `>>>` operator, in a similar way, is analogous to the monadic composition operator `>>=` and combines two arrows `arr a b` and `arr b c` by "wiring" the outputs of the first to the inputs to the second to get a new arrow `arr a c`. And lastly, the `first` operator, which takes the input arrow from `b` to `c` and converts it into an arrow on pairs with the second argument untouched, is also needed for actual useful code as without it, we wouldn't have a way to save input across arrows.

The most prominent instances of this interface are regular functions (`->`),

```

1 instance Arrow (->) where
2   arr f = f
3   f >>> g = g . f
4   first f = \ (a, c) -> (f a, c)

```

and the Kleisli type

```

1 data Kleisli m a b = Kleisli { run :: a -> m b }

```

as well:




Figure 1: syntactic sugar

```

1 instance Monad m => Arrow (Kleisli m) where
2   arr f = Kleisli $ return . f
3   f >>> g = Kleisli $ \a -> f a >>= g
4   first f = Kleisli $ \ (a,c) -> f a >>= \b -> return (b,c)

```

With this typeclass in place, Hughes also defined some syntactic sugar: The mirrored version of `first`, called `second`,

```

1 second :: Arrow arr => arr a b -> arr (c, a) (c, b)
2 second f = arr swap >>> first f >>> arr swap
3   where swap (x, y) = (y, x)

```

the `**` combinator which combines `first` and `second` to handle two inputs in one arrow,

```

1 (***) :: Arrow arr => arr a b -> arr c d -> arr (a, c) (b, d)
2 f *** g = first f >>> second g

```

and the `&&&` combinator that constructs an arrow which outputs 2 different values like `**`, but takes only one input.

```

1 (&&&) :: Arrow arr => arr a b -> arr a c -> arr (a, c) (b, d)
2 f &&& g = arr (\a -> (a, a)) >>> (f *** g)

```

A short example given by Hughes on how to use this is `add` over arrows:

```

1 add :: Arrow arr => arr a Int -> arr a Int -> arr a Int
2 add f g = (f &&& g) >>> arr \(u, v) -> u + v

```

The benefit of using the `Arrow` typeclass is that any type which is shown to be an arrow can be used in conjunction with this newly created `add` combinator.

Even though this example is quite simple, the power of the arrow interface immediately is clear: If a type is an arrow, it can automatically be used together with every library that works on arrows. Compared to simple monads, this enables us to write code that is more extensible, without touching the internals of the specific arrows.

*Note: In the definitions Hughes gave in his paper, the notation  $a \rightarrow b \rightarrow c$  for an arrow from  $b$  to  $c$  is used. We use the equivalent definition  $\text{arr } a \rightarrow b$  for an arrow from  $a$  to  $b$  instead, to make it easier to find the arrow type in type signatures.*



## 4 Utility Functions

Before we go into detail on parallel arrows, we introduce some utility combinators first, that will help us later: `map`, `foldl` and `zipWith` on arrows.

The `mapArr` combinator lifts any arrow `arr a b` to an arrow `arr [a] [b]` [8]:

```

1 mapArr :: ArrowChoice arr => arr a b -> arr [a] [b]
2 mapArr f =
3   arr listcase >>>
4   arr (const []) ||| (f *** mapArr f >>> arr (uncurry ()))
5   where
6     listcase [] = Left ()
7     listcase (x:xs) = Right (x,xs)

```

Similarly, we can also define `foldlArr` that lifts any arrow `arr (b, a) b` with a neutral element `b` to `arr [a] b`:

```

1 foldlArr :: (ArrowChoice arr, ArrowApply arr) => arr (b, a) b -> b -> arr [a] b
2 foldlArr f b =
3   arr listcase >>>
4   arr (const b) |||
5   ( first (arr (\a -> (b, a))) >>> f >>> arr (foldlArr f) >>> app)
6   where
7     listcase [] = Left []
8     listcase (x:xs) = Right (x,xs)

```

Finally, with the help of `mapArr`, we can define `zipWithArr`, which lifts any arrow `arr (a, b) c` to an arrow `arr ([a], [b]) [c]`.

```

1 zipWithArr :: ArrowChoice arr => arr (a, b) c -> arr ([a], [b]) [c]
2 zipWithArr f = (arr $ \ (as, bs) -> zipWith (,) as bs) >>> mapArr f

```

These combinators make use of the `ArrowChoice` typeclass, which allows us to use the `|||` combinator. It takes two arrows `arr a c` and `arr b c` and combines them into a new arrow `arr (Either a b) c` which pipes all `Left` `a`'s to the first arrow and all `Right` `b`'s to the second arrow.

```

1 (|||) :: ArrowChoice arr a c -> arr b c -> arr (Either a b) c

```

With the `zipWithArr` combinator we can also write a combinator `listApp`, which lifts a list of arrows `[arr a b]` to an arrow `arr [a] [b]`.

```


1 listApp :: (ArrowChoice arr, ArrowApply arr) => [arr a b] -> arr [a] [b]
2 listApp fs = (arr $ \ as -> (fs, as)) >>> zipWithArr app

```

This combinator also makes use of the `ArrowApply` typeclass which allows us to evaluate arrows with `app :: arr (arr a b, a) c`.

## 5 Parallel Arrows

We have seen what Arrows are and how they can be used as a general interface to computation. In the following section we will discuss how Arrows can also be looked at as a general interface not only to computation, but to **parallel computation** as well. We start by introducing the interface and explaining the reasonings behind it. Then, we discuss some implementations using existing parallel Haskell. Finally, we explain why using Arrows for expressing parallelism is beneficial.



### 5.1 The ArrowParallel typeclass

As we have seen earlier, in its purest form, parallel computation (on functions) can be looked at as the execution of some functions  $a \rightarrow b$  in parallel:

```
1 parEvalN :: [a -> b] -> [a] -> [b]
```

Translating this into arrow terms gives us a new operator `parEvalN` that lifts a list of arrows `[arr a b]` to a parallel arrow `arr [a] [b]` (This combinator is similar to our utility function `listApp`, but does parallel instead of serial evaluation).

```
1 parEvalN :: (Arrow arr) => [arr a b] -> arr [a] [b]
```

With this definition of `parEvalN`, parallel execution can be looked at as yet another arrow combinator. But as the implementation may differ depending on the actual type of the arrow `arr` and we want this to be an interface for different backends, we have to introduce the new typeclass `ArrowParallel` to host this combinator.

```
1 class Arrow arr => ArrowParallel arr a b where
2   parEvalN :: [arr a b] -> arr [a] [b]
```

Sometimes parallel Haskell require additional configuration parameters as information about the execution environment. This is why we also introduce an additional `conf` parameter to the function. We also do not want `conf` to be a fixed type, as the configuration parameters can differ for different instances of `ArrowParallel`. So we add it to the type signature of the typeclass as well.

```
1 class Arrow arr => ArrowParallel arr a b conf where
2   parEvalN :: conf -> [arr a b] -> arr [a] [b]
```

Note that we don't require the `conf` parameter in every implementation. If it is not needed, we usually want to allow the `conf` type parameter to be of any type and don't even evaluate it by blanking it in the type signature of the

implemented `parEvalN`, as we will see in the implementation of the Multicore and the ParMonad backend.

## 5.2 Multicore Haskell

The Multicore Haskell implementation of this class is straightforward using `listApp` from chapter 4 combined with the `using` operator from Multicore Haskell.

```

1 instance (NFData b, ArrowApply arr, ArrowChoice arr) =>
2   ArrowParallel arr a b conf where
3     parEvalN _ fs = listApp fs >>> arr (flip using $ parList rdeepseq)

```

We hardcode the `parList rdeepseq` strategy here, as in this context it is the only one making sense, since we usually want the output list to be fully evaluated to its normal form.

## 5.3 ParMonad

The ParMonad implementation makes use of Haskell's laziness and ParMonad's `spawnP :: NFData a => a -> Par (IVar a)` function, which forks away the computation of a value and returns an `IVar` containing the result in the Par monad.

We therefore apply each function to its corresponding input value with `app` and then fork the computation away with `arr spawnP` inside a `zipWithArr` call. This yields a list `[Par (IVar b)]`, which we then convert into `Par [IVar b]` with `arr sequenceA`. In order to wait for the computation to finish, we map over the `IVars` inside the ParMonad with `arr (>>= mapM get)`. The result of this operation is a `Par [b]` from which we can finally remove the monad again by running `arr runPar` to get our output of `[b]`.

```

1 instance (NFData b, ArrowApply arr, ArrowChoice arr) =>
2   ArrowParallel arr a b conf where
3     parEvalN _ fs =
4       (arr $ \as -> (fs, as)) >>>
5       zipWithArr (app >>> arr spawnP) >>>
6       arr sequenceA >>>
7       arr (>>= mapM get) >>>
8       arr runPar

```

## 5.4 Eden

For the Multicore and ParMonad implementation we could use general instances of `ArrowParallel` that just require the `ArrowApply` and `ArrowChoice` typeclasses. With Eden this is not the case as we can only spawn a list of functions and we cannot extract simple functions out of arrows. While we could still manage to have only one class in the module by introducing a typeclass like

```

1 class (Arrow arr) => ArrowUnwrap arr where
2   arr a b -> (a -> b)

```

, we don't do it in this paper, as this seems too hacky. For now, we just implement `ArrowParallel` for normal functions

```

1 instance (Trans a, Trans b) => ArrowParallel (->) a b conf where
2   parEvalN _ fs as = spawnF fs as

```

and the Kleisli type.

```

1 instance (Monad m, Trans a, Trans b, Trans (m b)) =>
2   ArrowParallel (Kleisli m) a b conf where
3   parEvalN conf fs =
4     (arr $ parEvalN conf (map (\(Kleisli f) -> f) fs)) >>>
5     (Kleisli $ sequence)

```

## 5.5 Benefits of parallel Arrows

Okay so?

We have seen that we can wrap parallel Haskell's inside of the `ArrowParallel` interface, but why do we abstract parallelism this way and what does this approach do better than the other parallel Haskell's?

- **Arrow API benefits:** With the `ArrowParallel` typeclass we do not lose any benefits of using arrows as `parEvalN` is just yet another arrow combinator. The resulting arrow can be used in the same way a potential serial version could be used. This is a big advantage of this approach, especially compared to the monad solutions as we do not introduce any new types. We can just "plug" in parallel parts into our sequential programs without having to change anything.
- **Abstraction:** With the `ArrowParallel` typeclass, we abstracted all parallel implementation logic away from the business logic. This gives us the beautiful situation of being able to write our code against the interface the typeclass gives us without being bound to any parallel Haskell. So as an example, during development, we can run the code on the simple Multicore version and afterwards deploy it on a cluster by converting it into an Eden version, by just swapping out the actual `ArrowParallel` instance.

## 6 Basic Skeletons

With the `ArrowParallel` typeclass in place and implemented, we can now implement some basic parallel skeletons.

### 6.1 `parEvalNLazy`




`parEvalN` is 100% strict, which means that it fully evaluates all passed arrows. Sometimes this might not be feasible, as it will not work on infinite lists of functions like e.g. `map (arr . (+)) [1..]` or just because we need the arrows evaluated in chunks. `parEvalNLazy` fixes this. It works by first chunking the input from `[a]` to `[[a]]` with the given `ChunkSize` in `arr` (`chunksOf chunkSize`). These chunks are then fed into a list `[arr [a] [b]]` of parallel arrows created by feeding chunks of the passed `ChunkSize` into the regular `parEvalN` by using `listApp`. The resulting `[[b]]` is lastly converted into `[b]` with `arr concat`.

```

1 parEvalNLazy :: (ArrowParallel arr a b conf, ArrowChoice arr, ArrowApply arr) =>
2   conf -> ChunkSize -> [arr a b] -> (arr [a] [b])
3 parEvalNLazy conf chunkSize fs =
4   arr (chunksOf chunkSize) >>>
5   listApp fchunks >>>
6   arr concat
7   where
8     fchunks = map (parEvalN conf) $ chunksOf chunkSize fs

```

### 6.2 `parEval2`



We have only talked about the parallelization arrows of the same type until now. But sometimes we want to parallelize heterogeneous types as well. However, we can implement such a `parEval2` combinator which combines two arrows `arr a b` and `arr c d` into a new parallel arrow `arr (a, c) (b, d)` quite easily with the help of the `ArrowChoice` typeclass. The idea is to use the `+++` combinator which combines two arrows `arr a b` and `arr c d` and transforms them into `arr (Either a c) (Either b d)` to get a common arrow type that we can then feed into `parEvalN`.

We start by transforming the  $(a, c)$  input into a 2-element list `[Either a c]` by first tagging the two inputs with **Left** and **Right** and wrapping the right element in a singleton list with **return** so that we can combine them with `arr (uncurry (:))`. Next, we feed this list into a parallel arrow running on 2 instances of `f +++ g` as described above. After the calculation is finished we convert the resulting `[Either b d]` into `([b], [d])` with `arr partitionEithers`. The two lists in this tuple contain only 1 element each by construction, so we can finally just convert the tuple to `(b, d)` in the last step.

```

1 parEval2 :: (ArrowChoice arr,
2   ArrowParallel arr (Either a c) (Either b d) conf) =>
3   conf -> arr a b -> arr c d -> arr (a, c) (b, d)
4 parEval2 conf f g =
5   arr Left *** (arr Right >>> arr return) >>>
6   arr (uncurry (:)) >>>
7   parEvalN conf (replicate 2 (f +++ g)) >>>
8   arr partitionEithers >>>
9   arr head *** arr head

```

## 7 Syntactic Sugar

For basic arrows, we have the `***` combinator which allows us to combine two arrows `arr a b` and `arr c d` into an arrow `arr (a, c) (b, d)` which does both computations at once. This can easily be translated into a parallel version with `parEval2`, but for this we require a backend which has an implementation that does not require any configuration (hence the `()` as the `conf` parameter in the following code snippet).

```
1 (|***|) :: (ArrowChoice arr, ArrowParallel arr (Either a c) (Either b d) ())) =>
2   arr a b -> arr c d -> arr (a, c) (b, d)
3 (|***|) = parEval2 ()
```

With this we can analogously to the serial `&&&` define the parallel `|&&&|`.

```
1 (|&&&|) :: (ArrowChoice arr, ArrowParallel arr (Either a a) (Either b c) ())) =>
2   arr a b -> arr a c -> arr a (b, c)
3 (|&&&|) f g = (arr $ \a -> (a, a)) >>> f |***| g
```

## 8 Futures

Consider the following arbitrary, but interesting, parallel arrow combinator:

```
1 someCombinator :: (Arrow arr) => [arr a b] -> [arr b c] -> arr [a] [c]
2 someCombinator fs1 fs2 = parEvalN () fs1 >>> rightRotate >>> parEvalN () fs2
```

In a distributed environment, the resulting arrow of this combinator first evaluates all `[arr a b]` in parallel, sends the results back to the master node, rotates the input once and then evaluates the `[arr b c]` in parallel to then gather the input once again on the master node. While this could be rewritten into only one `parEvalN` call by directly wiring the arrows properly together, this example illustrates an important problem: When using a `ArrowParallel` backend that resides on multiple computers, all communication between the nodes is done via the master node, as shown in the Eden trace in figure 2. This can become a serious bottleneck in heavy threaded applications. To fix this, we have to in-




Figure 2: Communication between 4 threads without Futures

roduce a way that enables the nodes to communicate directly with each other. Thankfully, Eden, the distributed parallel Haskell we have used in this paper so far, already ships with the concept of RD (remote data) that enables this behaviour. But as we want code written against our API to be implementation agnostic, we have to wrap this context. We do this with the `Future` typeclass:

```
1 class Future fut a | a -> fut where
2   put :: (Arrow arr) => arr a (fut a)
3   get :: (Arrow arr) => arr (fut a) a
```

As RD is only type synonym for communication type that Eden uses internally, we have to use some wrapper classes to fit that definition, though:

```
1 data RemoteData a = RD { rd :: RD a }
```



```

3 instance (Trans a) => Future RemoteData a where
4   put = arr (\a -> RD { rd = release a })
5   get = arr rd >>> arr fetch

```

For ParMonad and Multicore we can use a basic dummy wrapper because we have shared memory in a single node:

```

1 data BasicFuture a = BF { val :: a }
2
3 instance (NFData a) => Future BasicFuture a where
4   put = arr (\a -> BF { val = a })
5   get = arr val

```

To fit the ArrowParallel instances we gave earlier, we also have to give the necessary NFData and Trans instances - the latter only being needed in Eden. We need this implementation for our RemoteData wrapper

```

1 instance NFData (RemoteData a) where
2   rnf = rnf . rd
3 instance Trans (RemoteData a)

```

and the following for the BasicFuture dummy type:

```

1 instance (NFData a) => NFData (BasicFuture a) where
2   rnf = rnf . val

```

Going back to our communication example we can use this Future concept in order to enable direct communications between the nodes in the following way:

```

1 someCombinator :: (Arrow arr) => [arr a b] -> [arr b c] -> arr [a] [c]
2 someCombinator fs1 fs2 =
3   parEvalN () (map (>>> put) fs1) >>>
4   rightRotate >>>
5   parEvalN () (map (get >>>) fs2)

```

In a distributed environment, this gives us a communication scheme with messages going through the master node only if it is needed - similar to what is shown in the trace in fig. 3.





Figure 3: Communication between 4 threads with Futures

## 9 Map-based Skeletons

### 9.1 parMap



`parMap` is probably the most common skeleton for parallel programs. We can implement it with `ArrowParallel` by repeating an arrow `arr a b` and then passing it into `parEvalN` to get an arrow `arr [a] [b]`. Just like `parEvalN`, `parMap` is 100 % strict.

```

1 parMap :: (ArrowParallel arr a b conf) =>
2   conf -> (arr a b) -> (arr [a] [b])
3 parMap conf f = parEvalN conf (repeat f)

```

### 9.2 parMapStream




As `parMap` is 100% strict it has the same restrictions as `parEvalN` compared to `parEvalNLazy`. So it makes sense to also have a `parMapStream` which behaves like `parMap`, but uses `parEvalNLazy` instead of `parEvalN`.

```

1 parMapStream :: (ArrowParallel arr a b conf, ArrowChoice arr, ArrowApply arr) =>
2   conf -> ChunkSize -> arr a b -> arr [a] [b]
3 parMapStream conf chunkSize f = parEvalNLazy conf chunkSize (repeat f)

```

### 9.3 farm



`parMap` spawns every single computation in a new thread (at least for the instances of `ArrowParallel` we gave in this paper). This can be quite wasteful and a `farm` that equally distributes the workload over `numCores` workers (if `numCores` is greater than `actualProcessorCount`, the fastest processor(s) to finish will get more tasks) seems useful.

```

1 farm :: (ArrowParallel arr a b conf,
2   ArrowParallel arr [a] [b] conf, ArrowChoice arr) =>
3   conf -> NumCores -> arr a b -> arr [a] [b]
4 farm conf numCores f =
5   unshuffle numCores >>>
6   parEvalN conf (repeat (mapArr f)) >>>
7   shuffle

```

The definition of `unshuffle` is

```

1 unshuffle :: (Arrow arr) => Int -> arr [a] [[a]]
2 unshuffle n = arr (\xs -> [takeEach n (drop i xs) | i <- [0..n-1]])
3
4 takeEach :: Int -> [a] -> [a]
5 takeEach n [] = []
6 takeEach n (x:xs) = x : takeEach n (drop (n-1) xs)

```

, while `shuffle` is defined as:

```

1 shuffle :: (Arrow arr) => arr [[a]] [a]
2 shuffle = arr (concat . transpose)

```

(These were taken from Eden's source code [9] and translated into Arrows.)

### 9.4 farmChunk



As `farm` is basically just `parMap` with a different work distribution, it is, again, 100% strict. So we define `farmChunk` which uses `parEvalNLazy` instead of `parEvalN` like this:

```
1 farmChunk :: (ArrowParallel arr a b conf, ArrowParallel arr [a] [b] conf,  
2   ArrowChoice arr, ArrowApply arr) =>  
3   conf -> ChunkSize -> NumCores -> arr a b -> arr [a] [b]  
4 farmChunk conf chunkSize numCores f =  
5   unshuffle numCores >>>  
6   parEvalNLazy conf chunkSize (repeat (mapArr f)) >>>  
7   shuffle
```

## 9.5 parMapReduce

– this does not completely adhere to Google’s definition of Map Reduce as it – the mapping function does not allow for ”reordering” of the output – The original Google version can be found at <https://de.wikipedia.org/wiki/MapReduce>

```
1 parMapReduceDirect :: (ArrowParallel arr [a] b conf,  
2   ArrowApply arr, ArrowChoice arr) =>  
3   conf -> ChunkSize -> arr a b -> arr (b, b) b -> b -> arr [a] b  
4 parMapReduceDirect conf chunkSize mapfn foldfn neutral =  
5   arr (chunksOf chunkSize) >>>  
6   parMap conf (mapArr mapfn >>> foldlArr foldfn neutral) >>>  
7   foldlArr foldfn neutral
```

## 10 Topology Skeletons

Even though many algorithms can be expressed by parallel maps, some problems require more sophisticated skeletons. The Eden library leverages this problem and already comes with more predefined skeletons, among them are a `pipe`, `ring` and a `torus` implementation [10, 11]. These seem like reasonable candidates to be ported to our arrow based parallel Haskell to prove that we can express such skeletons with Arrows as well.

### 10.1 pipe

The parallel pipe skeleton is semantically equivalent to folding over a list `[arr a a]` of arrows with `>>>`, but does this in parallel, meaning that the arrows don't have to reside on the same thread/machine. We implement this skeleton using the `ArrowLoop` typeclass which enables us to use `loop :: arr (a, b) (c, b) -> arr a c`.

```
1 pipeSimple :: (ArrowLoop arr, ArrowParallel arr a a conf) =>
2   conf -> [arr a a] -> arr a a
3 pipeSimple conf fs =
4   loop (arr snd &&& (arr (uncurry (:)>>> lazy) >>>
5     parEvalN conf fs)) >>>
6   arr last
```

where `lazy` is defined as:

```
1 lazy :: (Arrow arr) => arr [a] [a]
2 lazy = arr (\ ~(x:xs) -> x : lazy xs)
```

However, using this definition directly, will result in the master node becoming a potential bottleneck in distributed environments as described in chapter 8. Therefore, a more sophisticated version that uses Futures internally is a good idea:

```
1 pipe :: (ArrowLoop arr, ArrowParallel arr (fut a) (fut a) conf,
2   Future fut a) =>
3   conf -> [arr a a] -> arr a a
4 pipe conf fs = unliftFut (pipeSimple conf (map liftFut fs))
```

Sometimes, this pipe definition can be a bit inconvenient, especially if we want to pipe arrows of mixed types together, i.e. `arr a b` and `arr b c`. By wrapping these two arrows inside a common type

```
1 pipe2 :: (ArrowLoop arr, ArrowChoice arr,
2   ArrowParallel arr (fut ([a], [b]), [c])) (fut ([a], [b]), [c])) conf,
3   Future fut ([a], [b]), [c])) =>
4   conf -> arr a b -> arr b c -> arr a c
5 pipe2 conf f g =
6   (arr return &&& arr (const [])) &&& arr (const []) >>>
7   pipe conf (replicate 2 (unify f g)) >>>
8   arr snd >>>
```

```

9  arr head
10  where
11  unify :: (ArrowChoice arr) =>
12    arr a b -> arr b c -> arr ([a], [b]), [c] ([a], [b]), [c]
13  unify f g =
14    (mapArr f *** mapArr g) *** arr (\_ -> []) >>>
15    arr (\((a, b), c) -> ((c, a), b))

```

Note that extensive use of this combinator over `pipe` with a hand-written combination data-type will probably result in worse performance because of more communication overhead from the many calls to `parEvalN`. Nonetheless, we can define a parallel piping operator `|>>>|` which is semantically equivalent to `>>>` similar to the other parallel syntactic sugar from chapter 7:

```

1  (|>>>|) :: (ArrowLoop arr, ArrowChoice arr,
2    ArrowParallel arr (fut ([a], [b]), [c])) (fut ([a], [b]), [c]) ()
3    Future fut ([a], [b]), [c]) =>
4    arr a b -> arr b c -> arr a c
5  (|>>>|) = pipe2 ()


```

## 10.2 ring

Eden's ring implementation allows a ring of functions to communicate over direct channels. We can rewrite its functionality easily again with the use of `loop`. In each loop we start by rotating the intermediary input from the nodes `[fut r]` with `second (rightRotate >>> lazy)`. We have to feed the intermediary input into `lazy` or the evaluation would hang. The reasoning was explained by [10]:

Note that the list of ring inputs `ringIns` is the same as the list of ring outputs `ringOuts` rotated by one element to the right using the auxiliary function `rightRotate`. Thus, the program would get stuck without the lazy pattern, because the ring input will only be produced after process creation and process creation will not occur without the first input.

Next, we zip the resulting `([i], [fut r])` to `[(i, fut r)]` with `arr (uncurry zip)` so we can feed that into a our input arrow `arr (i, r) (o, r)`, which we transform into `arr (i, fut r) (o, fut r)` before lifting it to `arr [(i, fut r)] [(o, fut r)]` to get a list `[(o, fut r)]`. Finally we unzip this list into `([o], [fut r])`. Plugging this arrow `arr ([i], [fut r]) ([o], fut r)` into the definition of `loop` from earlier gives us `arr [i] [o]`, our ring arrow.



```

1 ring :: (ArrowLoop arr, Future fut r,
2   ArrowParallel arr (i, fut r) (o, fut r) conf) =>
3   conf ->
4   arr (i, r) (o, r) ->
5   arr [i] [o]
6 ring conf f =
7   loop (second (rightRotate >>> lazy) >>>
8     arr (uncurry zip) >>>
9     parMap conf (second get >>> f >>> second put) >>>
10    arr unzip)

```

with rightRotate:

```

1 rightRotate :: (Arrow arr) => arr [a] [a]
2 rightRotate = arr $ \ list -> case
3   list of [] -> []
4   xs -> last xs : init xs

```

and lazy:


```

1 lazy :: (Arrow arr) => arr [a] [a]
2 lazy = arr (\ ~(x:xs) -> x : lazy xs

```

This combinator can be used for example to calculate the shortest paths in a graph using Warshall's algorithm. Further details on this can be found in [10].

### 10.3 torus



```

1 torus :: (ArrowLoop arr, ArrowChoice arr, ArrowApply arr,
2         ArrowParallel arr (c, fut [a], fut [b]) (d, fut [a], fut [b]) conf,
3         Future fut [a], Future fut [b]) =>
4         conf ->
5         arr (c, [a], [b]) (d, [a], [b]) ->
6         arr [[c]] [[d]]
7 torus conf f =
8   loop (second ((mapArr rightRotate >>> lazy) ***
9               (arr rightRotate >>> lazy)) >>>
10        arr (uncurry3 (zipWith3 lazyzip3)) >>>
11        (arr length >>> arr unshuffle) &&&
12        (shuffle >>> parEvalN conf (repeat (ptorus f))) >>>
13        app >>> arr (map unzip3) >>> arr unzip3 >>> threetotwo)
14
15 uncurry3 :: (a -> b -> c -> d) -> (a, (b, c)) -> d
16 uncurry3 f (a, (b, c)) = f a b c
17
18 lazyzip3 :: [a] -> [b] -> [c] -> [(a, b, c)]
19 lazyzip3 as bs cs = zip3 as (lazy bs) (lazy cs)
20
21 ptorus :: (Arrow arr, Future fut [a], Future fut [b]) =>
22         arr (c, [a], [b]) (d, [a], [b]) ->
23         arr (c, fut [a], fut [b]) (d, fut [a], fut [b])
24 ptorus f =
25   arr (\ ~(c, fas, fbs) -> (c, get fas, get fbs)) >>>
26   f >>>
27   arr (\ ~(c, as, bs) -> (c, put as, put bs))
28
29 threetotwo :: (Arrow arr) => arr (a, b, c) (a, (b, c))
30 threetotwo = arr $ \ ~(a, b, c) -> (a, (b, c))

```

In [10], matrix multiplication using the gentlemans algorithm was given as a example of using this skeleton. Adapting this nodefunction to our arrow API gives us:

```

1 nodefunction :: Int ->
2   ((Matrix, Matrix), [Matrix], [Matrix]) ->
3   ([Matrix], [Matrix], [Matrix])
4 nodefunction n ((bA, bB), rows, cols) =
5   ([bSum], bA:nextAs, bB:nextBs)
6   where bSum =
7     foldl' matAdd (matMult bA bB) (zipWith matMult nextAs nextBs)
8     nextAs = take (n-1) rows
9     nextBs = take (n-1) cols

```

Comparing the trace from a call using our arrow definition of the torus (fig. 4) with the Eden version (fig. 5) shows that the behaviour of the arrow version is comparable.







Figure 4: Matrix Multiplication with a torus (Parrows)

## 11 Benchmarks

To check the performance of our new parallel arrow API, we conducted some benchmarks. These are based on a sudoku solver from the examples for Simon Marlow's book "Parallel and Concurrent Programming in Haskell" which can be found on his GitHub [12]. The results are displayed in the following graphs which are ordered by problem size.

The Benchmarks were run on a Core i7-3970X CPU @ 3.5GHz with 6 cores and 12 threads. For sake of comparability with Simon Marlow's parallel version which uses the ParMonad, we use the ParMonad backend for the parallel arrow versions as well.



*Note: the bad result for parmonad-sudoku-parmap seems to be a artefact as it performs much better in the other benchmarks*






Figure 5: Matrix Multiplication with a torus (Eden)



As we can see, the parallel arrow versions of the program all have around the same speedup as the ParMonad based version. This means that using the `ArrowParallel` typeclass doesn't add any real notable overhead. Also, the slightly worse results for "parrows-sudoku-farm" and "parrows-sudoku-farmChunk" can

be explained by either imperfect parameters or by the fact that the benchmarks do not really require the scheduling introduced by the skeletons which in this case introduces unnecessary overhead. This would probably look different if the benchmark would use e.g. a divide and conquer approach to solve the sudoku puzzles.

## 12 Conclusion

As we have seen in this paper, arrows are a useful tool for composing parallel programs. By basing our parallel Haskell on them, we do not have to introduce new monadic types that wrap the computation and instead can use them just like they were regular sequential pure code. Performance-wise, parallel arrows are on par with existing parallel Haskells, as they do not introduce any notable overhead.

ArrowApply (or equivalent) are needed because we basically want to be able to produce intermediary results.

## References

- [1] John Hughes. Generalising monads to arrows. *Science of Computer Programming*, 37(1–3):67 – 111, 2000. ISSN 0167-6423. doi: [http://dx.doi.org/10.1016/S0167-6423\(99\)00023-4](http://dx.doi.org/10.1016/S0167-6423(99)00023-4). URL <http://www.sciencedirect.com/science/article/pii/S0167642399000234>.
- [2] Multicore’s parallel package on hackage. URL <https://hackage.haskell.org/package/parallel-3.2.1.0>. [Accessed on 01/12/2017].
- [3] Simon Marlow, Ryan Newton, and Simon Peyton Jones. A monad for deterministic parallelism. *SIGPLAN Not.*, 46(12):71–82, September 2011. ISSN 0362-1340. doi: 10.1145/2096148.2034685. URL <http://doi.acm.org/10.1145/2096148.2034685>.
- [4] Parmonad (control.monad.par) on hackage. URL <https://hackage.haskell.org/package/monad-par-0.3.4.8/>. [Accessed on 01/12/2017].
- [5] Eden modules on hackage, . URL <https://hackage.haskell.org/package/edenmodules-1.2.0.0/>. [Accessed on 01/12/2017].
- [6] Rita Loogen. *Eden – Parallel Functional Programming with Haskell*, pages 142–206. Springer Berlin Heidelberg, Berlin, Heidelberg, 2012. ISBN 978-3-642-32096-5. doi: 10.1007/978-3-642-32096-5\_4. URL [http://dx.doi.org/10.1007/978-3-642-32096-5\\_4](http://dx.doi.org/10.1007/978-3-642-32096-5_4).
- [7] Eden homepage, . URL <http://www.mathematik.uni-marburg.de/~eden/>. [Accessed on 01/12/2017].
- [8] John Hughes. *Programming with Arrows*, pages 73–129. Springer Berlin Heidelberg, Berlin, Heidelberg, 2005. ISBN 978-3-540-31872-9. doi: 10.1007/11546382\_2. URL [http://dx.doi.org/10.1007/11546382\\_2](http://dx.doi.org/10.1007/11546382_2).
- [9] Eden skeletons’ control.parallel.eden.map package source code, . URL <https://hackage.haskell.org/package/edenskel-2.1.0.0/docs/src/Control-Parallel-Eden-Map.html>. [Accessed on 02/12/2017].
- [10] Rita Loogen. Eden - parallel functional programming with haskell. URL <http://www.mathematik.uni-marburg.de/~eden/paper/edenCEFP.pdf>. [Accessed on 03/17/2017].
- [11] Eden skeletons’ control.parallel.eden.topology package source code, . URL <https://hackage.haskell.org/package/edenskel-2.0.0.2/docs/Control-Parallel-Eden-Topology.html>. [Accessed on 03/17/2017].
- [12] Simon Marlow. Sample code for “Parallel and Concurrent Programming in Haskell”. URL <https://github.com/simonmar/parconc-examples>. [Accessed on 01/12/2017].